Setting some parameters

In[472]:=
```
ClearAll[R, Mn, Edeut, Vzero]
ϵ = N[10^-4];
hbarc = 198;
R = 1.93 / hbarc;
Mn = 940.;
Edeut = 2.225;
```

Solving transcendental equation to get the square well height in MeV

In[478]:=
```
Vzero = Vzero /. FindRoot[
    Cot[Sqrt[Mn (Vzero - Edeut)] R] == -Sqrt[Edeut / (Vzero - Edeut)], {Vzero, 20}]
```

Out[478]=
```
38.9882
```

Defining the square well potential

In[479]:=
```
Vtotal[r_] = If[r < R, -Vzero, 0];
potplot = Plot[Vtotal[r], {r, 0, 3 R}, PlotStyle → Red]
```

Out[480]=



Solving the Schrodinger equation numerically for the bound state

In[496]:=

```
normali = 0.75 10^4;
(* Normalization factor for the wavefunction *)
crap = NDSolve[{(-1/Mn u''[r] + (Vtotal[r] + Edeut) u[r]) == 0,
   u[ϵ] == ϵ, u'[ϵ] == 1}, u, {r, ϵ, 5 R}];
BSplot = Plot[Evaluate[normali (u[r] /. crap)],
  {r, ϵ, 5 R}, Frame → True, PlotStyle → Green]
```

Out[497]=



Scattering states: Finding the scattering length

Let' s first plot the wavefunction at zero energy

In[489]:=
```
k = 0;
normali = 0.75 10^4;
(* Normalization factor for the wavefunction *)
crap = NDSolve[{(-1/Mn u''[r] + (Vtotal[r] - k^2) u[r]) == 0,
   u[ϵ] == ϵ, u'[ϵ] == 1}, u, {r, ϵ, 20 R}];
scattplot = Plot[Evaluate[normali (u[r] /. crap)],
  {r, ϵ, 5 R}, Frame → True, PlotStyle → Blue]
```

Out[491]=



Now let' s get the scattering length using method described in class

In[492]:=
```
ClearAll[crapA, rtest, Deriv, atest, ascatt];
ascatt := Module[{atest},
  crapA = NDSolve[{(-1/Mn u''[r] + Vtotal[r] u[r]) == 0,
     u[ϵ] == ϵ, u'[ϵ] == 1}, u, {r, ϵ, 20 R}];
  rtest = 4 R;
  Deriv = D[(u[r] /. crapA), r] /. r → rtest;
  atest = (rtest - ((u[r] /. crapA) /. r → rtest) / Deriv)]
```

Recall that experimentally a = 5.4 fm

In[494]:=
```
ascatt[[1]] hbarc
```

Out[494]=
```
5.26039
```

Combined plot

In[495]:=
```
Show[scattplot, BSplot, potplot]
```

Out[495]=



The nucleons in the deuteron live most of the time outside of the range of the potential that binds them!